
NAG Library Routine Document

H02CCF

Note: before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold italicised terms and
other implementation-dependent details.

1 Purpose

To supply optional parameters to H02CBF from an external file.

2 Specification

SUBROUTINE H02CCF (IOPTNS, INFORM)

INTEGER IOPTNS, INFORM

3 Description

H02CCF may be used to supply values for optional parameters to H02CBF. H02CCF reads an external
file and each line of the file defines a single optional parameter. It is only necessary to supply values for
those parameters whose values are to be different from their default values.

Each optional parameter is defined by a single character string of up to 72 characters, consisting of one or
more items. The items associated with a given option must be separated by spaces, or equal signs ¼½ �.
Alphabetic characters may be upper or lower case. The string

Print level = 1

is an example of a string used to set an optional parameter. For each option the string contains one or
more of the following items:

– a mandatory keyword;

– a phrase that qualifies the keyword;

– a number that specifies an integer or real value. Such numbers may be up to 16 contiguous characters
in Fortran 77’s I, F, E or D formats, terminated by a space if this is not the last item on the line.

Blank strings and comments are ignored. A comment begins with an asterisk (*) and all subsequent
characters in the string are regarded as part of the comment.

The file containing the options must start with Begin and must finish with End. An example of a valid
options file is:

Begin * Example options file
Print level = 10

End

Normally each line of the file is printed as it is read, on the current advisory message unit (see X04ABF),
but printing may be suppressed using the keyword Nolist. To suppress printing of Begin, Nolist must be
the first option supplied as in the file:

Begin
Nolist
Print level = 10

End

Printing will automatically be turned on again after a call to H02CBF and may be turned on again at any
time by you by using the keyword List.

Optional parameter settings are preserved following a call to H02CBF, and so the keyword Defaults is
provided to allow you to reset all the optional parameters to their default values prior to a subsequent call
to H02CBF.

A complete list of optional parameters, their abbreviations, synonyms and default values is given in
Section 11 in H02CBF.

4 References

None.

5 Parameters

1: IOPTNS – INTEGER Input

On entry: the unit number of the options file to be read.

Constraint: 0 � IOPTNS � 99.

2: INFORM – INTEGER Output

On exit: contains zero if the options file has been successfully read and a value > 0 otherwise, as
indicated below.

INFORM ¼ 1
IOPTNS is not in the range 0; 99½ �.

INFORM ¼ 2
Begin was found, but end-of-file was found before End was found.

INFORM ¼ 3
end-of-file was found before Begin was found.

6 Error Indicators and Warnings

If a line is not recognized as a valid option, then a warning message is output on the current advisory
message unit (see X04ABF).

7 Accuracy

Not applicable.

8 Further Comments

H02CDF may also be used to supply optional parameters to H02CBF. Note that if E04NFF/E04NFA is
used in the same program as H02CBF, then in general H02CCF will also affect the options used by
E04NFF/E04NFA.

9 Example

This example solves the same problem as the example for H02CBF, but in addition illustrates the use of
H02CCF and H02CDF to set optional parameters for H02CBF.

In this example the options file read by H02CCF is appended to the data file for the program (see
Section 9.2). It would usually be more convenient in practice to keep the data file and the options file
separate.

9.1 Program Text

Program h02ccfe

! H02CCF Example Program Text

! Mark 24 Release. NAG Copyright 2012.

! .. Use Statements ..

Use nag_library, Only: e04nfu, h02cbf, h02cbu, h02ccf, h02cdf, nag_wp, &
x04abf, x04acf, x04baf

! .. Implicit None Statement ..
Implicit None

! .. Parameters ..
Integer, Parameter :: iset = 1, lintvr = 1, mdepth = 30, &

nin = 5, ninopt = 7, nout = 6
Character (*), Parameter :: fname = ’h02ccfe.opt’

! .. Local Scalars ..
Real (Kind=nag_wp) :: obj
Integer :: i, ifail, inform, j, lda, ldh, &

liwork, lwork, mode, n, nclin, &
outchn, strtgy

Character (80) :: rec
! .. Local Arrays ..

Real (Kind=nag_wp), Allocatable :: a(:,:), ax(:), bl(:), bu(:), &
clamda(:), cvec(:), h(:,:), work(:), &
x(:)

Integer, Allocatable :: intvar(:), istate(:), iwork(:)
! .. Executable Statements ..

Write (rec,99996) ’H02CCF Example Program Results’
Call x04baf(nout,rec)

! Skip heading in data file
Read (nin,*)

Read (nin,*) n, nclin
lda = nclin
ldh = n

liwork = 2*n + 3 + 2*mdepth

! LWRK for default problem-type QP2

If (nclin==0) Then
lwork = n**2 + 9*n + 4*mdepth

Else
lwork = 2*n**2 + 9*n + 5*nclin + 4*mdepth

End If

Allocate (a(lda,n),ax(nclin),bl(n+nclin),bu(n+nclin),clamda(n+nclin), &
cvec(n),h(ldh,n),x(n+nclin),intvar(lintvr),istate(n+nclin), &
iwork(liwork),work(lwork))

Read (nin,*)(cvec(i),i=1,n)
Read (nin,*)((a(i,j),j=1,n),i=1,nclin)
Read (nin,*)(bl(i),i=1,n+nclin)
Read (nin,*)(bu(i),i=1,n+nclin)
Read (nin,*)(x(i),i=1,n)
Read (nin,*)((h(i,j),j=1,n),i=1,n)

! Set four options using H02CDF

Call h02cdf(’ Print Level = 1 ’)

Call h02cdf(’ Check Frequency = 10 ’)

Call h02cdf(’ Crash Tolerance = 0.05 ’)

Call h02cdf(’ Infinite Bound Size = 1.0D+25 ’)

! Set the unit number for advisory messages to OUTCHN

outchn = nout

Call x04abf(iset,outchn)

! Open the options file for reading

mode = 0

ifail = 0
Call x04acf(ninopt,fname,mode,ifail)

! Read the options file for the remaining options

Call h02ccf(ninopt,inform)

If (inform/=0) Then
Write (rec,99997) ’H02CCF terminated with INFORM = ’, inform
Call x04baf(nout,rec)
Go To 100

End If

strtgy = 2
intvar(1) = 4

Call h02cdf(’Nolist’)

Call h02cdf(’Print Level = 0’)

! Solve the problem

ifail = 0
Call h02cbf(n,nclin,a,lda,bl,bu,cvec,h,ldh,e04nfu,intvar,lintvr,mdepth, &

istate,x,obj,ax,clamda,strtgy,iwork,liwork,work,lwork,h02cbu,ifail)

! Print out the best integer solution found

Write (rec,’()’)
Call x04baf(nout,rec)
Write (rec,’()’)
Call x04baf(nout,rec)
Write (rec,99999) obj
Call x04baf(nout,rec)
Call x04baf(nout,’ Components are ’)

Do i = 1, n
Write (rec,99998) i, x(i)
Call x04baf(nout,rec)

End Do

100 Continue

99999 Format (1X,’Optimal Integer Value is = ’,E20.8)
99998 Format (1X,’X(’,I3,’) = ’,F15.8)
99997 Format (A,I5)
99996 Format (1X,A)

End Program h02ccfe

9.2 Program Data

Begin Example options file for H02CCF
Feasibility Phase Iteration Limit = 5 * (Default = 70)
Optimality Phase Iteration Limit = 10 * (Default = 70)

End

H02CCF Example Program Data
7 7 :Values of N and NCLIN

-0.02 -0.20 -0.20 -0.20 -0.20 0.04 0.04 :End of CVEC
1.00 1.00 1.00 1.00 1.00 1.00 1.00
0.15 0.04 0.02 0.04 0.02 0.01 0.03
0.03 0.05 0.08 0.02 0.06 0.01 0.00
0.02 0.04 0.01 0.02 0.02 0.00 0.00
0.02 0.03 0.00 0.00 0.01 0.00 0.00
0.70 0.75 0.80 0.75 0.80 0.97 0.00
0.02 0.06 0.08 0.12 0.02 0.01 0.97 :End of matrix A

-0.01 -0.10 -0.01 -0.04 -0.10 -0.01 -0.01
-0.13 -1.0D+25 -1.0D+25 -1.0D+25 -1.0D+25 -9.92D-02 -3.0D-03 :End of BL
0.01 0.15 0.03 0.02 0.05 1.0D+25 1.0D+25

-0.13 -4.9D-03 -6.4D-03 -3.7D-03 -1.2D-03 1.0D+25 2.0D-03 :End of BU
-0.01 -0.03 0.00 -0.01 -0.10 0.02 0.01 :End of X
2.00 0.00 0.00 0.00 0.00 0.00 0.00
0.00 2.00 0.00 0.00 0.00 0.00 0.00
0.00 0.00 2.00 2.00 0.00 0.00 0.00
0.00 0.00 2.00 2.00 0.00 0.00 0.00
0.00 0.00 0.00 0.00 2.00 0.00 0.00
0.00 0.00 0.00 0.00 0.00 -2.00 -2.00
0.00 0.00 0.00 0.00 0.00 -2.00 -2.00 :End of matrix H

9.3 Program Results

H02CCF Example Program Results

Calls to H02CDF
---------------

Print Level = 1
Check Frequency = 10
Crash Tolerance = 0.05
Infinite Bound Size = 1.0D+25

OPTIONS file
------------

Begin Example options file for H02CCF
Feasibility Phase Iteration Limit = 5 * (Default = 70)
Optimality Phase Iteration Limit = 10 * (Default = 70)

End

Optimal Integer Value is = 0.37469662E-01
Components are
X( 1) = -0.01000000
X( 2) = -0.07332830
X( 3) = -0.00025809
X( 4) = 0.00000000
X( 5) = -0.06335433
X( 6) = 0.01410944
X( 7) = 0.00283128

___________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________________

